

An Adventure Game For The Bally Arcade and Commodore Vic 20



Cavern Quest

The Bally Arcade (or Astrocade) is probably the most undersung of the home computer systems, and undeservedly so. The color graphics and sound capabilities were, and in many cases still are, years ahead of the competition, both in the cartridge games and with Bally Basic. The most serious limitations have been the small, calculator-style keyboard and the mere 1800 bytes of RAM.

Was I crazy then, to attempt to write an adventure game for the Bally? Perhaps. But I tried, and I think you will enjoy the result.

I knew that no magic was strong enough to fit any kind of traditional Adventure game into those 1800 bytes. In addition, a novice at the Bally keypad could take days to type in "Take the Amulet," so I elected to use joystick 1 for input.

The scenario of the game is straightforward. Jesse James, the Robin Hood or Attila the Hun of Missouri (depending on your point of view), has hidden a treasure deep in a cave, and you want to go in, find it, and bring it out. That's it. No magic birds, sword-wielding ogres or dissembling oracles. Not only would that have been quite a task in 1800 bytes, but it turns out that a simple, realistic exploration of a three-dimensional cave is difficult enough, thank you.

The cavern passages do not form a traditional maze, as there is often more than one path to the treasure, and any path is liable to loop back on itself—sometimes in an inexplicable way. Make a map.

Bruce Jaeger, 1253 Ashland Ave., St. Paul, MN 55104.

Bruce Jaeger

The Program

The first third of the program generates the passages of the cave, using the random number generator, so unlike most adventure games, the cave is different each time. Also at this time the various artifacts of the James boys are strewn about, and the treasure hidden. It takes about half a minute or so for the computer to do its work.

The rest of the program is the playing section. The player enters the cave, and is told whether passages lead north, south, east, west, up or down. Pushing the joystick forward is the equivalent of saying Go North, and the other compass directions correspond.

As you will never get both an Up and a Down choice at the same point, a simple pull of the trigger (or firing button) suffices to indicate climb or descent. Then it is simply a matter of wandering through the cave, keeping track of where you are, finding the treasure, and making your way out again.

To make the cavern as large as possible none of the fine Bally sound effects or graphics other than color changes have been used. Because of the restricted memory, it is also important not to type in any spaces between commands, except as noted in lines 54 and 55.


Vic Version

I have included a translation of Cavern

Quest for the Vic computer (or any other regular Basic computer, with modifications), as Bally Basic code is often difficult to translate. The cavern can be made much larger if you wish, by increasing the value of variable B in line 15 to as much as your available memory will allow.

Again, I have used no sound effects or fancy graphics, to make translation of the program into other dialects of Basic easier. The POKEs in lines 15 and 260 change the Vic screen and border colors, and the joystick subroutine at line 1000 is obviously peculiar to Vic. For Pet or Commodore 64, substitute the keyboard subroutine at 1000. For other microcomputers, simply write your own subroutine that returns V=1 for north, V=-1 for south, J=1 for east, J=-1 for west, and R=1 for up or down.

The TIS in lines 240 and 1005 refers to the built-in real-time clock. This is not a necessary part of the game, but it is interesting to see how long you have been lost! □



Cavern Quest, continued...

Cavern Quest for Bally.

```
1 BC=48;FC=125;CLEAR ;E=10;Z=0;B=300;T=B;C=B/2;FOR A=1TO B+11;@ (A)=Z;NEXT A
2 A=E;@ (A)=1;NT=0
3 D=RND (7);CLEAR ;PRINT T;T=T-1;IF T=0GOTO 24
4 IFD=2D=-1
5 IFD=3D=-E
6 IFD=4D=E
7 IFD=5D=-C
8 IFD=6D=C
9 IFD=7GOTO 18
10 H=A+D;IF H<1GOTO 3
11 IFH>BGOTO 3
12 K=@ (H);IF K=3GOTO 3
13 IF K=4GOTO 3
14 IF K>BGOTO 3
15 @ (H)=1;IF D=-C@ (A)=3;@ (H)=4
16 IF D=C@ (H)=3;@ (A)=4
17 A=H;GOTO 3
18 IF RND (3)#2GOTO 10
19 IF @ (A)#1GOTO 3
20 Z=Z+1;K=0;IF Z>EGOTO 3
21 Y=RND (E);K=K+1;IF K=40GOTO 3
22 IF @ (Y+B)#0GOTO 21
23 @ (A)=Y+B;@ (Y+B)=1;GOTO 3
24 @ (A)=B+11;CLEAR ;A=E;G=0
25 CLEAR ;IF A#EGOTO 27
26 PRINT "ENTRANCE";IF G=1BC=102;FC=16;PRINT "## YOU MADE IT! ##";;NT=2;GTO
27 PRINT "PASSAGES LEAD:";IF A>E;IF @ (A-E)PRINT"NORTH"
28 IF A<(B-E)IF @ (A+E)PRINT "SOUTH"
29 IF @ (A-1)PRINT "WEST"
30 IF A<BIF @ (A+1)PRINT "EAST"
31 IF @ (A)=4PRINT "UP"
32 IF @ (A)=3PRINT "DOWN"
33 PRINT ;IF @ (A)<BGOTO46
34 PRINT "YOU SEE";X=@ (A)-B
35 IF X=1PRINT "A BOOT"
36 IF X=2PRINT "A SKELETON"
37 IF X=3PRINT "A SHOVEL"
38 IF X=4PRINT "MILDEWED SADDLEBAGS"
39 IF X=5PRINT "A RUSTY PISTOL"
40 IF X=6PRINT "A BROKEN LANTERN"
41 IF X=7PRINT "'J. JAMES WAS HERE'"
42 IF X=8PRINT "YUCK! BATS!"
43 IF X=9PRINT "A BUSTED LADDER"
44 IF X=10PRINT "AN EMPTY TRUNK"
45 IF X=11PRINT "## TREASURE ##";G=1;@ (A)=1
46 V=0;J=0;R=0;J=JX(1);IF J#0GOTO 50
47 V=JY(1);IF V#0GOTO 50
48 R=TR(1);IF R#0GOTO 50
49 GOTO 46
50 IF V=1L=A-E
51 IF V=-1L=A+E
52 IF J=1L=A+1
53 IF J=-1L=A-1
54 IF @ (A)=3IF R L=A-C;GOTO56 ← Put a space
55 IF @ (A)=4IF R L=A+C ← after the 'R'
56 IF L<1GOTO 60 in lines
57 IF L>BGOTO 60 54 & 55!
58 IF @ (L)=0GOTO 60
59 A=L;GOTO 25
60 PRINT "##SOLID ROCK##";GOTO 46
```

